The analytical solution to the primary current distribution problem in ***Model 1*** obtained in the transformed coordinates using conformal mapping, along with the solution procedure, is given below.

First the original square domain in the $Z = X + IY$ coordinate system is transformed to a line from $-\infty$ to $+\infty$ in the w coordinate system using

$$\frac{dZ}{dw} = \frac{-I\,K[1]}{\sqrt{w}\sqrt{w-1}\sqrt{w-1-a}\sqrt{w-2-a}} \tag{1}$$

where K[1] and a are solved using the conditions $w = 1$ when $Z = 1 + 0\,I$ and $w = 1 + a$ when $Z = 1 + 1\,I$, corresponding to $X = 1, Y = 0$ and $X = 1, Y = 1$ in the original geometry, respectively. The value of a is found to be $\sqrt{2} - 1$. Now w is mapped to $Z2 = X2 + IY2$ using the transformation,

$$\frac{dZ2}{dw} = \frac{-I\,K[2]}{\sqrt{w}\sqrt{w-\frac{1}{\sqrt{2}}}\sqrt{w-1-a}\sqrt{w-2-a}} \tag{2}$$

where $Z2 = 1 + 0\,I$ when $w = \frac{1}{\sqrt{2}}$ is used to determine K[2], and $Z2 = 1 + H\,I$ when $w = 1 + a$ is used to obtain H. The final solution in the Z2 coordinate system is

$$\phi = 1 - \frac{\text{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right) Y2}{\text{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)} \tag{3}$$

```
>
> restart;
> Digits:=15;
```

$$Digits := 15$$

The domain is tranformed from Z = X+IY to w. The points tranformed are

```
> Zdomain:=[[0,0],[1,0],[1,1],[0,1]];
```

$$Zdomain := [[0,0],[1,0],[1,1],[0,1]]$$

```
> wdomain:=[0,1,1+a,a+2];
```

$$wdomain := [0,1,1+a,a+2]$$

```
> eq1:=diff(Z(w),w)=-I*K1/sqrt(w)/sqrt(w-1)/sqrt(w-a-1)/sqrt(w-a-2);
```

$$eq1 := \frac{d}{dw}\mathrm{Z}(w) = \frac{-I\,K1}{\sqrt{w}\,\sqrt{w-1}\,\sqrt{w-a-1}\,\sqrt{w-a-2}}$$

a is not known apriori. The value of a should be found to make sure [1,1] in the Z coordinate is transformed to [1+a] in the w coordinate.

```
> a:=sqrt(2)-1;
```

$$a := \sqrt{2}-1$$

Value of K1 is found using the transformation of [1,0] to 1 in the w coordinate

```
> eq11:=1=int(rhs(eq1),w=0..1);
```

$$eq11 := 1 = -K1\,\sqrt{2}\,\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)$$

```
> K1:=solve(eq11,K1);
```

$$K1 := -\frac{1}{2}\,\frac{\sqrt{2}}{\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)}$$

The height Y in the Z coordinate is found by integrating from 1 to 1+a in the w coordinate.

```
> simplify(int(rhs(eq1),w=1..1+a));
```

$$\frac{2\,I\,\sqrt{2}\,\mathrm{EllipticK}\left(\frac{\sqrt{2}-1}{\sqrt{2}+1}\right)}{\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)(\sqrt{2}+1)}$$

```
> evalf(%);
```

$$1.00000000000000\,I$$

The choice of a = sqrt(2)-1 gives the height of 1 for Z coordinate.

```
> eval(simplify(int(rhs(eq1),w=0..1+a)));
```

$$\frac{\sqrt{2}\,\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)+\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)+2\,I\,\sqrt{2}\,\mathrm{EllipticK}\left(\frac{\sqrt{2}-1}{\sqrt{2}+1}\right)}{\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)(\sqrt{2}+1)}$$

```
> evalf(%);
```

$$1.00000000000000 + 1.00000000000001\,I$$

integration from 0 to 1+a in the w coordinate gives 1,1 in the Z coordinate.

```
> simplify(int(rhs(eq1),w=0..1/sqrt(2)));
```

$$\frac{\mathrm{EllipticF}\left(\frac{\sqrt{2}}{\sqrt{2+\sqrt{2}}},\frac{\sqrt{2}}{2}\right)}{\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)}$$

```
> evalf(%);
```

$$0.500000000000001$$

Integrating w from 0 to wmid =1/sqrt(2) gives the point 0.5,0 in the Z coordinate

```
> wmid:=1/sqrt(2);
```

$$wmid := \frac{\sqrt{2}}{2}$$

Next w domain is transformed to Z2 domain Z2 = X2+IY2

```
> Z2domain:=[[0,0],[1,0],[1,H],[0,H]];
```

$$Z2domain := [[0,0],[1,0],[1,H],[0,H]]$$

```
> wdomain2:=[0,1/sqrt(2),1+a,a+2];
```

$$wdomain2 := \left[0,\frac{\sqrt{2}}{2},\sqrt{2},\sqrt{2}+1\right]$$

```
> eq2:=diff(Z2(w),w)=-I*K2/sqrt(w)/sqrt(w-wmid)/sqrt(w-1-a)/sqrt(w-a
  -2);
```

$$eq2 := \frac{d}{dw}\mathrm{Z2}(w) = \frac{-2\,I\,K2}{\sqrt{w}\,\sqrt{4\,w-2\sqrt{2}}\,\sqrt{w-\sqrt{2}}\,\sqrt{w-\sqrt{2}-1}}$$

K2 is found based on the transformation of wmid to [1,0] in Z2 coordinate.

```
> eq21:=1=int(rhs(eq2),w=0..wmid);
```

$$eq21 := 1 = -\frac{2\,K2\,\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}{\sqrt{\sqrt{2}+1}}$$

```
> K2:=solve(eq21,K2);
```

$$K2 := -\frac{1}{2}\frac{\sqrt{\sqrt{2}+1}}{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}$$

The corner 0,1 in the Z coordinate is mapped by integrating eq2 from 0 to 1 in the w coordinate

```
> int(rhs(eq2),w=0..1);
```

$$1+\frac{\mathrm{EllipticF}\left(\sqrt{2-\sqrt{2}},\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)I}{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}$$

```
> corner:=evalf(%);
```

$$corner := 1. + 0.559419351518322\, I$$

This is the point 1,0 in the original coordinate.

The height in the Z2 coorinate is found by integrating eq2 from wmid to 1+a.

```
> ytot:=int(rhs(eq2),w=wmid..1+a);
```

$$ytot := \frac{\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)I}{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}$$

```
> evalf(%);
```

$$1.22004159128347\, I$$

The magnitude in the Y direction is given by the coefficient of I, the imaginary number

```
> Ytot:=coeff(ytot,I);
```

$$Ytot := \frac{\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)}{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}$$

The analytial solution in the Z2 corordinate is a line in Y2 to satisfy simple zero flux conditions at X2 = 0 and X2 = 1.

```
> phianal:=1+b*Y2;
```

$$phianal := 1 + b\, Y2$$

The value of phi is zero at Y2 = Ytot (originally the cathode domain in the Z domain)

```
> bc:=subs(Y2=Ytot,phianal)=0;
```

$$bc := 1 + \frac{b\,\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)}{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)} = 0$$

```
> b:=solve(bc,b);
```

$$b := -\frac{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}{\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)}$$

The analytical solution is given by

```
> phianal;
```

$$1 - \frac{\mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right) Y2}{\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)}$$

```
> evalf(phianal);
```

$$1. - 0.819644188480505\ Y2$$

The potential at the corner is given by substituting the imaginary value of corner for Y2 in phinanal)

```
> phicorner:=subs(Y2=Im(corner),phianal);
```

$$phicorner := 1 - \frac{0.559419351518322\ \mathrm{EllipticK}\left(\frac{1}{\sqrt{2+\sqrt{2}}}\right)}{\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)}$$

```
> evalf(phicorner);
```

$$0.541475179604475$$

local flux/current density calculation, written in terms of w is

```
> curr:=b*rhs(eq2)/rhs(eq1);
```

$$curr := -\frac{\sqrt{\sqrt{2}+1}\,\sqrt{2}\,\mathrm{EllipticK}\left(\frac{\sqrt{2}}{2}\right)\sqrt{w-1}}{\mathrm{EllipticK}\left(\sqrt{\frac{\sqrt{2}+1}{2+\sqrt{2}}}\right)\sqrt{4w-2\sqrt{2}}}$$

average flux/current density calculation for the anode

```
> currave:=int((curr),w=0..wmid)/wmid;
```

$$currave := -\frac{1}{2} \frac{\sqrt{\sqrt{2}+1}\,(\sqrt{2}-1)\,\text{EllipticK}\left(\frac{\sqrt{2}}{2}\right)\left(2^{(3/4)}+2^{(1/4)}+\text{arctanh}\left(\frac{2^{(3/4)}}{2}\right)\right)\sqrt{2}}{\text{EllipticK}\left(\frac{2^{(3/4)}}{2}\right)}$$

```
> Digits:=25:
```

The average current density at Y =0, local current density at X = 0,Y=0 and potential at X=1,Y=0 (Corner) can be used to study convergence of FEM and other numerical methods

```
> evalf(currave),evalf(subs(w=0,curr)),evalf(phicorner);
```

-1.656507648777793388522396, -1.161311530233258689567781,
0.5414751796044734741869534

```
>
```